
This document lists all changes needed to correct problems in the TTCN implementation of test case 12.2a which is part of the ‘IMS_34229_IWD_14wk24’ test suite in ATS ‘iwd-TTCN3-B2013-03_D14wk24’.

Change 1 – f_TC_12_2a_IMS
	Function name
	 f_TC_12_2a_IMS

	Reason for change
	Test case calls f_IMS_PTC_ImsInfo_Reset to re-initialize miscellaneous state information, including the value of the branch parameter. This leads to re-start from the front of the available values for the branch parameter. However, RFC 3261 states the following: “The branch parameter value MUST be unique across space and time for all requests sent by the UA.”  

	Summary of change
	Feed the latest used value of the branch parameter into f_IMS_PTC_ImsInfo_Reset such that the next value can be taken from the pool of values rather than starting from the beginning again.

	TTCN module
	 IMS/12/IMS_CC_CallControlTestcases

	MCC160 Comment
	


Before change

    function f_TC_12_2a_IMS() runs on IMS_PTC

  {

    var template (present) MessageHeader v_MessageHeader_Ack;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;  /* @sic R5w140112: IMS_InviteRequestWithSdp_Type instead of IMS_DATA_REQ */

    var INVITE_Request v_InviteRequest;

    var template (present) SipUrl v_RequestSipUrl := f_SIP_BuildSipUri_RX(px_IMS_CalleeUri);

    f_IMS_CC_Preamble(IPCAN_SpeechCall, IMS_REGISTERED, IMS_Security);  // prose specifies only Annex C.2 => not applicable for GIBA

    f_IMS_TestBody_Set(true);

    // Step 1 - Make the UE attempt an IMS speech call

    // @siclog "Step 1" siclog@

    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall);

    // Step 2. Receive INVITE Request

    // @siclog "Step 2" siclog@

    v_InviteRequestWithSdp := f_IMS_MOCallSetup_AnnexC21_Step2_A4();

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_InviteRequestWithSdp.RoutingInfo);

    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    // Step 3. SS sends 504 Server Time-out.

    // @siclog "Step 3" siclog@

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine504,

                                                                   f_IMS_CommonResponse_504_MessageHeaderTX(v_InviteRequest.msgHeader),

                                                                   cs_MessageBody_XML(cs_TIMS3GPP_Restoration))));

    // Step 4. Receive ACK

    // @siclog "Step 4" siclog@

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine504);

    IMS_Server.receive(car_IMS_Ack_Request(cr_ACK_Request(v_RequestSipUrl, v_MessageHeader_Ack)));

    f_IMS_CC_AbortCall();    // to abort call establishment procedure

    f_IMS_RegisterInfo_Release();  // release registration information at IPCAN

    f_IMS_PTC_ImsInfo_Reset();     // reset ImsInfo

    f_IMS_PTC_Security_Release();  // release security

    // Steps 5 to 12. UE performs initial registration Annex C.2

    // @siclog "Step 5 - 12" siclog@

    f_IMS_Registration();

    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_MO_SpeechCall);

  }
After change

     function f_TC_12_2a_IMS() runs on IMS_PTC

  {

    var template (present) MessageHeader v_MessageHeader_Ack;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;  /* @sic R5w140112: IMS_InviteRequestWithSdp_Type instead of IMS_DATA_REQ */

    var INVITE_Request v_InviteRequest;

    var template (present) SipUrl v_RequestSipUrl := f_SIP_BuildSipUri_RX(px_IMS_CalleeUri);

    f_IMS_CC_Preamble(IPCAN_SpeechCall, IMS_REGISTERED, IMS_Security);  // prose specifies only Annex C.2 => not applicable for GIBA

    f_IMS_TestBody_Set(true);

    // Step 1 - Make the UE attempt an IMS speech call

    // @siclog "Step 1" siclog@

    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall);

    // Step 2. Receive INVITE Request

    // @siclog "Step 2" siclog@

    v_InviteRequestWithSdp := f_IMS_MOCallSetup_AnnexC21_Step2_A4();

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_InviteRequestWithSdp.RoutingInfo);

    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    // Step 3. SS sends 504 Server Time-out.

    // @siclog "Step 3" siclog@

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine504,

                                                                   f_IMS_CommonResponse_504_MessageHeaderTX(v_InviteRequest.msgHeader),

                                                                   cs_MessageBody_XML(cs_TIMS3GPP_Restoration))));

    // Step 4. Receive ACK

    // @siclog "Step 4" siclog@

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine504);

    IMS_Server.receive(car_IMS_Ack_Request(cr_ACK_Request(v_RequestSipUrl, v_MessageHeader_Ack)));

    f_IMS_CC_AbortCall();    // to abort call establishment procedure

    f_IMS_RegisterInfo_Release();  // release registration information at IPCAN

    f_IMS_PTC_ImsInfo_Reset(vc_IMS_Global.ImsInfo.BranchIndex);     // reset ImsInfo

    f_IMS_PTC_Security_Release();  // release security

    // Steps 5 to 12. UE performs initial registration Annex C.2

    // @siclog "Step 5 - 12" siclog@

    f_IMS_Registration();

    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_MO_SpeechCall);

  }
Change 2 – f_IMS_PTC_ImsInfo_Reset
	Function name
	 f_IMS_PTC_ImsInfo_Reset

	Reason for change
	 See Change 1

	Summary of change
	Forward the received value for the branch parameter and forward it to the initializing template

	TTCN module
	 Common/IMS/IMS_Component

	MCC160 Comment
	


Before change

    function f_IMS_PTC_ImsInfo_Reset() runs on IMS_PTC

  {

    vc_IMS_Global.ImsInfo := valueof(cs_IMS_ImsInfo_Init(vc_IMS_Global.ImsInfo.RanType,

                                                         vc_IMS_Global.ImsInfo.UE_Release,

                                                         vc_IMS_Global.ImsInfo.Pcscf,

                                                         vc_IMS_Global.ImsInfo.HomeDomainName,

                                                         vc_IMS_Global.ImsInfo.PrivateUserId,

                                                         vc_IMS_Global.ImsInfo.SecurityScheme));

  }
After change

     function f_IMS_PTC_ImsInfo_Reset(integer branchIndex := 0) runs on IMS_PTC

  {

    vc_IMS_Global.ImsInfo := valueof(cs_IMS_ImsInfo_Init(vc_IMS_Global.ImsInfo.RanType,

                                                         vc_IMS_Global.ImsInfo.UE_Release,

                                                         vc_IMS_Global.ImsInfo.Pcscf,

                                                         vc_IMS_Global.ImsInfo.HomeDomainName,

                                                         vc_IMS_Global.ImsInfo.PrivateUserId,

                                                         vc_IMS_Global.ImsInfo.SecurityScheme,

                                                         branchIndex));

  }
Change 3 – cs_IMS_ImsInfo_Init
	Template name
	 cs_IMS_ImsInfo_Init

	Reason for change
	 See Change 1

	Summary of change
	Use the received value to populate the branch parameter in ImsInfo data structure. For the case of no value explicitly given, use the value of 0 (just like in current code).

	TTCN module
	 Common/IMS/IMS_Component

	MCC160 Comment
	


Before change

    template (value) IMS_PTC_ImsInfo_Type cs_IMS_ImsInfo_Init(template (omit) IPCAN_RAN_Type p_RanType,

                                                            template (omit) integer p_UE_Release,

                                                            charstring p_Pcscf,

                                                            charstring p_HomeDomainName := px_IMS_HomeDomainName,

                                                            charstring p_PrivateUserId := px_IMS_Private_UserId,

                                                            IMS_SecurityScheme_Type p_SecurityScheme) :=

  { /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT) */

    IMPUList := {px_IMS_PublicUserIdentity1, px_IMS_PublicUserIdentity2, px_IMS_PublicUserIdentity3},

    UE_Address := omit,

    NW_Address := omit,

    TransportProtocol := udp,

    RanType := p_RanType,

    UE_Release := p_UE_Release,

    SigCompStarted := false,

    HomeDomainName := p_HomeDomainName,

    PrivateUserId := p_PrivateUserId,

    ContactUrl := omit,

    BranchIndex := 0,
    CallIdCnt := 0,

    Pcscf := p_Pcscf,

    Registration := cs_IMS_RegistrationInfo_Init,

    Dialog := fl_IMS_PTC_DialogList_Init(),

    SecurityScheme := p_SecurityScheme

  };
After change

     template (value) IMS_PTC_ImsInfo_Type cs_IMS_ImsInfo_Init(template (omit) IPCAN_RAN_Type p_RanType,

                                                            template (omit) integer p_UE_Release,

                                                            charstring p_Pcscf,

                                                            charstring p_HomeDomainName := px_IMS_HomeDomainName,

                                                            charstring p_PrivateUserId := px_IMS_Private_UserId,

                                                            IMS_SecurityScheme_Type p_SecurityScheme,

                                                            integer branchIndex := 0) :=

  { /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT) */

    IMPUList := {px_IMS_PublicUserIdentity1, px_IMS_PublicUserIdentity2, px_IMS_PublicUserIdentity3},

    UE_Address := omit,

    NW_Address := omit,

    TransportProtocol := udp,

    RanType := p_RanType,

    UE_Release := p_UE_Release,

    SigCompStarted := false,

    HomeDomainName := p_HomeDomainName,

    PrivateUserId := p_PrivateUserId,

    ContactUrl := omit,

    BranchIndex := branchIndex,
    CallIdCnt := 0,

    Pcscf := p_Pcscf,

    Registration := cs_IMS_RegistrationInfo_Init,

    Dialog := fl_IMS_PTC_DialogList_Init(),

    SecurityScheme := p_SecurityScheme

  };
