
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of IMS emergency call testcase 15.1 which is part of the IMS test suite in the ‘iwd-TTCN3-B2013-03_D14wk37’ delivery.

The test case can be demonstrated to run with one LTE UE supporting IMS. Execution logs are provided as evidence.

3. Verification Test Summary

Test Case:
TC_15_14
Test Group:
IMS_IWD_14wk37
ATS Version:
iwd-TTCN3-B2013-03_D14wk37
System Simulator used:
Anritsu Protocol Conformance Test System ME7832L
UE used:
Qualcomm MSM8974
Verification Status:
PASS


4. Corrections required for IMS test case 15_14
This section describes the TTCN changes required to ‘iwd-TTCN3-B2013-03_D14wk37’ release
Change 1
	Function name
	fl_Simservs_GetSpecificService()

	Reason for change
	 As specified in change 5, to check the ‘rule-deactivated’ 3rd parameter to function 
fl_SupplementaryService_CheckActivated() needs to be passed

	Summary of change
	Defined the flag v_RuleDeactivation and passed to fl_SupplementaryService_CheckActivated()

	TTCN module
	IMS\15\IMS_CC_SSXcapTestcases.ttcn


Before change

function fl_Simservs_GetSpecificService(Simservs p_Simservs,

                                          SS_Type p_SupplementaryService) return SupplementaryServiceInfoList_Type

  {

    var template (value) SupplementaryServiceInfoList_Type v_ServiceList := {};

    var Ruleset v_Ruleset := fl_Ruleset(omit);

    var AbsService_group v_AbsService;

    var boolean v_Active;

    select (p_SupplementaryService) {

      case (SS_CLIP) {

        v_AbsService := fl_Simservs_GetService(p_Simservs, cr_Originating_identity_presentation, cs_Originating_identity_presentation);

        v_Active := match(v_AbsService, cr_Originating_identity_presentation(true));

        v_ServiceList[0] := cs_SupplementaryService_Info(v_AbsService, v_Active, v_Ruleset);

      }

      case (SS_CLIR) {

        v_AbsService := fl_Simservs_GetService(p_Simservs, cr_Originating_identity_presentation_restriction, cs_Originating_identity_presentation_restriction);

        v_Active := match(v_AbsService, cr_Originating_identity_presentation_restriction(true));

        v_ServiceList[0] := cs_SupplementaryService_Info(v_AbsService, v_Active, v_Ruleset);

      }

      case (SS_COLP) {

        v_AbsService := fl_Simservs_GetService(p_Simservs, cr_Terminating_identity_presentation, cs_Terminating_identity_presentation);

        v_Active := match(v_AbsService, cr_Terminating_identity_presentation(true));

        v_ServiceList[0] := cs_SupplementaryService_Info(v_AbsService, v_Active, v_Ruleset);

      }

      case (SS_TIR) {

        v_AbsService := fl_Simservs_GetService(p_Simservs, cr_Terminating_identity_presentation_restriction, cs_Terminating_identity_presentation_restriction);

        v_Active := match(v_AbsService, cr_Terminating_identity_presentation_restriction(true));

        v_ServiceList[0] := cs_SupplementaryService_Info(v_AbsService, v_Active, v_Ruleset);

      }

      case (SS_CFU,

            SS_CFNRy,

            SS_CFB,

            SS_CFNL,

            SS_CFNRc) {

        v_AbsService := fl_Simservs_GetService(p_Simservs, cr_Communication_diversion, cs_Communication_diversion);

        v_Active := match(v_AbsService, cr_Communication_diversion(true));

        v_Ruleset := fl_Ruleset(v_AbsService.communication_diversion.ruleset);

        v_ServiceList[0] := cs_SupplementaryService_Info(v_AbsService, v_Active, v_Ruleset);

      }

      case (SS_BAIC,

            SS_BAIC_ANONYMOUS,

            SS_BAIC_EXCEPT_SPECIFIC_USER) {

        v_AbsService := fl_Simservs_GetService(p_Simservs, cr_Incoming_communication_barring, cs_Incoming_communication_barring);

        v_Active := match(v_AbsService, cr_Incoming_communication_barring(true));

        v_Ruleset := fl_Ruleset(v_AbsService.incoming_communication_barring.ruleset);

        v_ServiceList[0] := cs_SupplementaryService_Info(v_AbsService, v_Active, v_Ruleset);

      }

      case (SS_BARRING_IN_AND_OUT_ROAM) {

        v_AbsService := fl_Simservs_GetService(p_Simservs, cr_Incoming_communication_barring, cs_Incoming_communication_barring);

        v_Active := match(v_AbsService, cr_Incoming_communication_barring(true));

        v_Ruleset := fl_Ruleset(v_AbsService.incoming_communication_barring.ruleset);

        v_ServiceList[0] := cs_SupplementaryService_Info(v_AbsService, v_Active, v_Ruleset);

        v_AbsService := fl_Simservs_GetService(p_Simservs, cr_Outgoing_communication_barring, cs_Outgoing_communication_barring);

        v_Active := match(v_AbsService, cr_Outgoing_communication_barring(true));

        v_Ruleset := fl_Ruleset(v_AbsService.outgoing_communication_barring.ruleset);

        v_ServiceList[1] := cs_SupplementaryService_Info(v_AbsService, v_Active, v_Ruleset);

      }

      case else {

        FatalError(__FILE__, __LINE__, "unknown supplementatry service");

      }

    }

    return valueof(v_ServiceList);

  }
After change
  function fl_XCAP_SupplementaryService_CommonTestBody(SS_Type p_SupplementaryService,

                                                       charstring p_PublicUserIdentity) runs on IMS_XCAP_PTC

  {

    const boolean tsc_Activated := true;

    var float v_MaxDuration := 300.0;       /* @sic R5s130794 change 21: increased to 300.0 sic@ */

    var Simservs v_SimservsDocument;

    var SupplementaryServiceInfoList_Type v_ServiceList;

    var boolean v_RuleDeactivation := false;
    f_IMS_XCAP_InitialiseSuplementaryServices();

    f_IMS_TestBody_Set(true);

    // Step 1 - Activate the service

    f_UT_IMS_ActivateDeactivateSupplementaryService(MMI, true, p_PublicUserIdentity, p_SupplementaryService);

    // Step 2 - HTTP Request / Response exchange

    f_IMS_HTTP_Transaction(initial, v_MaxDuration);

    v_SimservsDocument := f_IMS_XCAP_GetDocument();

    v_ServiceList := fl_Simservs_GetSpecificService(v_SimservsDocument, p_SupplementaryService);

    if (not fl_SupplementaryService_CheckActivated(v_ServiceList, tsc_Activated,v_RuleDeactivation)) {

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 2 - supplementary service is not activated");

    }

    // check ruleset for communication forwarding, communication barring

    select (p_SupplementaryService) {

      case (SS_CFU,

            SS_CFNRy,

            SS_CFB,

            SS_CFNL,

            SS_CFNRc) {

        if (not fl_CommunicationForwarding_CheckRuleset(p_SupplementaryService, v_ServiceList[0].ruleset)) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 2 - supplementary service: invalid rule set for communication forwarding");

        }

         v_RuleDeactivation := true;

      }

      case (SS_BAIC_ANONYMOUS,

            SS_BAIC_EXCEPT_SPECIFIC_USER) {

        if (not fl_IncomingCommunicationBarring_CheckRuleset(p_SupplementaryService, v_ServiceList[0].ruleset)) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 2 - supplementary service: invalid rule set for incoming communication barring");

        }

     v_RuleDeactivation := true;      
  }

      case (SS_BARRING_IN_AND_OUT_ROAM) {

        if (not fl_CommunicationBarringWhileRoaming_CheckRuleset(v_ServiceList[0].ruleset)) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 2 - supplementary service: invalid rule set for incoming communication barring");

        }

        /*if (not fl_CommunicationBarringWhileRoaming_CheckRuleset(v_ServiceList[1].ruleset)) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 2 - supplementary service: invalid rule set for outgoing communication barring");

        }*/

      }

    }

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    // Step 3 - Deactivate the service

    f_UT_IMS_ActivateDeactivateSupplementaryService(MMI, false, p_PublicUserIdentity, p_SupplementaryService);

    // Step 4

    f_IMS_HTTP_Transaction(subsequent, v_MaxDuration);

    v_SimservsDocument := f_IMS_XCAP_GetDocument();

    v_ServiceList := fl_Simservs_GetSpecificService(v_SimservsDocument, p_SupplementaryService);

    if (not fl_SupplementaryService_CheckActivated(v_ServiceList, not tsc_Activated, v_RuleDeactivation)) {

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4 - supplementary service is still activated");

    }

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    f_IMS_TestBody_Set(false);

  }

Change 2
	Function name
	f_TC_15_14_IMS()

	Reason for change
	As the UE has two PDNs, during the switchoff procedure, UE can send PDN DISCONNECT followed by detach procedure. Hence, second PDN is removed before moving into switchoff procedure like the case of emergency calls

	Summary of change
	As defined in change 7 of R5s141051, to deactivate the second bearer, IPCAN_XCAP_Signalling sent

	TTCN module
	IMS\15\IMS_CC_SSXcapTestcases.ttcn


Before change

  function f_TC_15_14_IMS() runs on IMS_XCAP_PTC

  {

    var charstring v_XcapUserId := fl_XCAP_Preamble();

    var SS_Type v_SupplementaryService := SS_BAIC_ANONYMOUS;

    // @siclog "Step 1 - 4" siclog@

    fl_XCAP_SupplementaryService_CommonTestBody(v_SupplementaryService, v_XcapUserId);

    f_IMS_CC_Postamble(IPCAN_MO_IMS_Signalling);

  }
After change

…

  function f_TC_15_14_IMS() runs on IMS_XCAP_PTC

  {

    var charstring v_XcapUserId := fl_XCAP_Preamble();

    var SS_Type v_SupplementaryService := SS_BAIC_ANONYMOUS;

    // @siclog "Step 1 - 4" siclog@

    fl_XCAP_SupplementaryService_CommonTestBody(v_SupplementaryService, v_XcapUserId);

    f_IMS_CC_Postamble(IPCAN_XCAP_Signalling);

  }
Change 3

	Function name
	fl_Rule_CheckCondition

	Reason for change
	As specified in change 4, instead of matching the conditions along with the xml format. Decvlaue can be invoked to decode and compare

	Summary of change
	fl_match_condition invoked as per change 4

	Source of change
	IMS\15\IMS_CC_SSXcapTestcases.ttcn


Before change:

	  function fl_Rule_CheckCondition(RuleType p_Rule,

                                  template (omit) XSD.String p_Condition) return boolean

  {

    var XSD.String v_Element;

    var XSD.String v_Condition;

    var XSD.String v_PatternString1;

    var XSD.String v_PatternString2;

    var template (present) XSD.String v_PatternTemplate;

    var integer i;

    var integer k;

 // remove below line

   if (not ispresent(p_Condition)) {  // => "unconditional"

   //if (true) { 

      if (match(p_Rule.conditions, cr_ConditionsType_Empty ifpresent)) {     // @sic R5s130794 change 26 sic@

        return true;

      }

    } else {

      v_Condition := valueof(p_Condition);

      if (ispresent(p_Rule.conditions)) {

        for (i := 0; i < lengthof(p_Rule.conditions.choice_list); i := i + 1) {

          if (ischosen(p_Rule.conditions.choice_list[i].elem_list)) {

            for (k := 0; k < lengthof(p_Rule.conditions.choice_list[i].elem_list); k := k + 1) {

              v_Element := p_Rule.conditions.choice_list[i].elem_list[k];

              v_PatternString1 := "<" & v_Condition & "/>";

              v_PatternString2 := "<" & v_Condition & "></" & v_Condition & ">";

              v_PatternTemplate := (pattern v_PatternString1, pattern v_PatternString2);

              if (match(v_Element, v_PatternTemplate)) {

                return true;
              }

            }

          }

        }

      }

    }

    return false;

  }


After change:

	function fl_Rule_CheckCondition(RuleType p_Rule,

                                  template (omit) XSD.String p_Condition) return boolean

  {

    var XSD.String v_Element;

    var XSD.String v_Condition;

    var XSD.String v_PatternString1;

    var XSD.String v_PatternString2;

    var template (present) XSD.String v_PatternTemplate;

    var integer i;

    var integer k;

 // remove below line

   if (not ispresent(p_Condition)) {  // => "unconditional"

   //if (true) { 

      if (match(p_Rule.conditions, cr_ConditionsType_Empty ifpresent)) {     // @sic R5s130794 change 26 sic@

        return true;

      }

    } else {

      v_Condition := valueof(p_Condition);

      if (ispresent(p_Rule.conditions)) {

        for (i := 0; i < lengthof(p_Rule.conditions.choice_list); i := i + 1) {

          if (ischosen(p_Rule.conditions.choice_list[i].elem_list)) {

            for (k := 0; k < lengthof(p_Rule.conditions.choice_list[i].elem_list); k := k + 1) {

              if (fl_MatchCondition(p_Rule.conditions.choice_list[i].elem_list[k], v_Condition)) {  /* @sic R5s140975 - Additional MCC160 changes sic@ */

                return true;

              }

            }

          }

        }

      }

    }

    return false;

  }    


Change 4

	Function name
	fl_MatchCondition

	Reason for change
	Instead of using<> during the comparision, decvalue can be called to get the value of it so that it can be compared



	Summary of change
	Decvalue invoked to decode the xml fragments

	Source of change
	IMS\15\IMS_CC_SSXcapTestcases.ttcn


	//----------------------------------------------------------------------------

  function fl_MatchCondition(XSD.String p_ConditionsElem,

                             charstring p_GivenCondition) return boolean

  {

    var bitstring v_Bitstring := oct2bit(char2oct(p_ConditionsElem));

    var Anonymous v_Anonymous;

    var Roaming v_Roaming;

    var No_answer v_No_answer;

    var Busy v_Busy;

    var Not_registered v_Not_registered;

    var Not_reachable v_Not_reachable;

    var Rule_deactivated v_Rule_deactivated;

    var integer v_DecvalueResult := -1;

    select (p_GivenCondition) {

      case ("anonymous")      {v_DecvalueResult := decvalue(v_Bitstring, v_Anonymous);}

      case ("roaming")        {v_DecvalueResult := decvalue(v_Bitstring, v_Roaming);}

      case ("no-answer")      {v_DecvalueResult := decvalue(v_Bitstring, v_No_answer);}

      case ("busy")           {v_DecvalueResult := decvalue(v_Bitstring, v_Busy);}

      case ("not-registered") {v_DecvalueResult := decvalue(v_Bitstring, v_Not_registered);}

      case ("not-reachable")  {v_DecvalueResult := decvalue(v_Bitstring, v_Not_reachable);}

      case ("rule-deactivated") {v_DecvalueResult := decvalue(v_Bitstring, v_Rule_deactivated);}

      case else {FatalError(__FILE__, __LINE__, "Invalid condition");}

    }

    return (v_DecvalueResult == 0);

  }



Change 5

	Function name
	fl_SupplementaryService_CheckActivated

	Reason for change
	UE can also send ‘rule-deactivated’ element to the network to deactivate the corresponding rule. So that it can be reactivated later. In this case, active flag of the corresponding serive will be still true, but, ‘rule-deactivated’ will be present during the deactivation procedure invoked. Hence, incase of, ‘active’ flag of the service is true, then, ;rule-deactivated’ needs to be checked.

#draft ProseCR will be raised for next RAN5 meeting



	Summary of change
	fl_Rule_CheckCondition is invoked inside fl_SupplementaryService_CheckActivated when service active flag is true

	Source of change
	IMS\15\IMS_CC_SSXcapTestcases.ttcn


Before change:

	function fl_SupplementaryService_CheckActivated(SupplementaryServiceInfoList_Type p_ServiceList,

                                                  boolean p_Activated) return boolean

  {

    var integer i;

    for (i:=0; i < lengthof(p_ServiceList); i:=i+1) {

      if (p_ServiceList[i].active != p_Activated) {

        return false;

      }

    }

    return true;

  }


After change:

	function fl_SupplementaryService_CheckActivated(SupplementaryServiceInfoList_Type p_ServiceList,

                                                  boolean p_Activated,

                                                  boolean p_CheckRuleDeactivation := false) return boolean

  {

    var integer i,j;

    var RuleType v_Rule;
    for (i:=0; i < lengthof(p_ServiceList); i:=i+1) {

      if (p_ServiceList[i].active != p_Activated) {

         if(p_CheckRuleDeactivation) {

              for (j := 0; j < lengthof(p_ServiceList[i].ruleset.rule_list); j := j + 1) {

               v_Rule := p_ServiceList[i].ruleset.rule_list[j];

                 if (fl_Rule_CheckCondition(v_Rule, "rule-deactivated")) {

                   return true;

                 }

              }

          }
        return false;

      }

    }

    return true;

  }
    


Change 6 - 14
Changes from 1 to 9 of TTCN CR R5s141051

5. Execution Log Files
Qualcomm UE
The Qualcomm MSM8974 passed this test case on Anritsu ME7832L LTE System in LTE band 1. The documentation below is enclosed as evidence of the successful test case run [1]:

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

6 References

	[1]
	R5s141114    Supporting information for agreement of IMS TC 15.14 in LTE FDD mode.
                          This archive comprises:

                        -  html and text format execution log files

